


3 מחשבת

תיכנות
לשעות
הפנאי

zx spectrum



משהו על חברת "מחשבת" – מערכות למידה

זו חברה, אשר אחד מתחומי התמחותה העיקריים הוא פיתוח מערכות לימוד עצמי.

מחברי יחידות לימוד אלה עוסקים במשך שנים בפיתוח מערכות ללימוד עצמי עבור המגזר האזרחי והצבאי.

חל איסור מוחלט להעתיק או לשכפל יחידת לימוד זו בשלמותה או בחלקה לכל מטרה שהיא או לעשות בה שימוש מסחרי כלשהו, ללא רשות בכתב מאת חברת מחשבת – מ.ל. בע"מ.

© 1985 – כל הזכויות שמורות למחשבת – מ.ל. בע"מ.

ת.ד. 48032 ת"א מיקוד 61480. טל. 03-233872

הוצאה לאור – מחשבת מ.ל. בע"מ.

**Copyright © 1985 by MAKH-SHEVET M.L. Ltd.
P.O.Box 48032 Tel-Aviv 61480, Israel**

ZX Spectrum

BUG

MICROCOMPUTERS BOOKS & SOFTWARE

המרכז לספרות עזר ותאגיד למחשבים

מחשבת

צעדים ראשונים במחשב
בלימוד עצמי

יחידה 3

פיתוח משחקי מחשב – טלויזיה

תוכן העניינים

5	מבוא
6	פרק א: יורים ברעים
17	פרק ב: עפרון וידיאו
26	פרק ג: מרוצי סוסים
37	פרק ד: חפש את המטמון
40	סיום
41	תשובות

מבוא

הגעת ליחידת הלימוד השלישית והאחרונה בקורס. כבר עכשיו אתה נושא מטען ידע לא מבוטל אשר מאפשר לך לכתוב תכניות המציירות צורות גרפיות מענינות, תוך שימוש בצבעים ובצלילים של המחשב. כמו כן, אנו מקווים כי יש לך הרגשה שאתה מתחיל להשתלט על המחשב. את היחידה השלישית אנו מקדישים לפיתוח משחקי מחשב. תוך כדי פיתוח המשחקים תלמד עוד כמה פקודות ומושגים יסודיים הקשורים לכתיבת תכניות. בסיומה של יחידת לימוד זו יהיו בידך מספיק כלים כדי לפתח תכניות שתהיה מעוניין בהן.

באיחולי המשך לימוד

מרתק ומהנה


דני קדם ויצחק קליסקי

מחשבת - מערכות למידה

פרק א


יורים ברעים!

מי מאתנו לא שיחק לפחות פעם אחת באחד מאלפי משחקי-
הטלויזיה?
הפעם, לשם שינוי, אתה עומד לבנות משחק כזה:




חללית "הרעים" (>) חולפת ללא הרף
לרוחב המסך משמאל לימין. החללית
שלך (\$) אורבת לה למטה...

1



ברגע שנדמה לך ש"הרעים" (>)
נמצאים בדיוק מעל לחללית שלך,
אתה לוחץ על אחד המקשים של המחשב...

2



ברגע זה נעצרת חללית "הרעים"
והטיל שירית (*) מתחיל לעלות
מן החללית שלך ישר למעלה!

3

מטרת המשחק: לפגוע בחללית האויב (>)!
בוא נבדוק מה אתה כבר יודע לתכנת כדי לבנות את המשחק
הזה, ומה עדיין עליך ללמוד:

- האם אתה יודע כיצד להניע את הרעים לרוחב המסך? (כן/לא)
- האם אתה יודע לצייר את החללית שלך בתחתית המשחק? (כן/לא)
- האם אתה יודע להניע את ה"טיל" מן החללית עד למעלה? (כן/לא)
- האם אתה יודע לתכנת את המחשב כך, שידע מתי בדיוק
לחצת על אחד המקשים כדי לירות על הטיל? (כן/לא)

?

האמת היא, שרק לשאלה האחרונה התשובה היא "לא"!
 אתה עדיין אינך יודע כיצד לכתוב פקודות כאלה, שיגרמו
 לטיל "לעוף" למעלה רק ברגע שתלחץ על מקש כלשהו!

לשם כך עליך ללמוד 2 פקודות חדשות. הראשונה שבהן:
IN KEY\$ (אין-קי דולר) - המקש הלחץ

כתוב את התוכנית הבאה לתוך המחשב (אחרי שמחקת כל תוכנית

קודמת - NEW):

```
10 PRINT IN KEY $; 000000000000
20 GO TO 10         000000000000
                    000000000000
                    000000000000
```

שים לב: IN KEY \$ כתוב בירוק מעל המקש.
 כדי לקבל אותו עליך לעשות אותן פעולות
 כפי שעשית כדי לקבל CHR\$ (ראה יחידה 2 עמוד 32).

הרץ את התוכנית (RUN). לחץ על מקשים שונים. בדוק איך

התוכנית מגיבה ללחיצות שלך.

אם לא שכחת לכתוב ; (נקודה-פסיק) בסוף שורה 10, הרי
 שהמחשב הדפיס על פני המסך את המקש עליו לחצת באותו רגע!
 איך זה קורה? מה סודה של התוכנית הקצרה הזאת?
 הסוד של התוכנית נעוץ ב-IN KEY \$ (אין-קי דולר):

כשהמחשב מגיע בתוכנית ל-IN KEY \$ הוא קורא איזה מקש
 לחוץ באותו רגע, וממשיך בביצוע התוכנית.

ואם אין מקש לחוץ באותו רגע?
במקרה כזה המחשב מתעלם מ-\$ IN KEY וממשיך הלאה בביצוע התוכנית.

אם כך, מה מבצע המחשב בשורה 10:

10 PRINT IN KEY \$;

הוא מדפיס את המקש הלחוץ באותו רגע - וממשיך הלאה לשורה 20:

20 GO TO 10

?
כשאינך לוחץ על המקשים כלל - האם המחשב עוצר ומחכה בשורה 10, או שהוא ממשיך ומתרוצץ בין הפקודות בתוך לולאה אינסופית?

תשובה 1 בעמוד 41

?
מדוע התוכנית מדפיסה מספר פעמים אותו סימן, אם אתה ממשיך ללחוץ על המקש?

תשובה 2 בעמוד 41

?
רוצה להשתכנע שהמחשב מתרוצץ בלולאה אינסופית גם כשאינך לוחץ על המקשים?

פקוד על המחשב להשמיע צפצוף קצר בכל פעם שהוא עובר משורה 10 לשורה 20.

תשובה 3 בעמוד 41

אתה בודאי שומע שהמחשב מצפצף גם כשאינך לוחץ על מקש כלשהו - דבר זה מוכיח שהוא "מתרוצץ" ומבצע את התוכנית שוב ושוב גם כשאינך לוחץ על המקשים

כתוב תוכנית שתגרום למחשב להדפיס במרכז המסך (15, 10) למשל) את המקש עליו אתה לוחץ:

10 _____
20 _____

תשובה 4 בעמוד 41

פקודת תנאי

וכעת נפנה לפקודה השנייה הדרושה לך כדי לתכנת את "ההדק" של משחק הטלויזיה:

... THEN ... IF ... (אִי־פִּי ... דִּן ...) - אִם ... - אִז...

התוכנית הקודמת עדיין רשומה במחשב שלך:


```
10 PRINT AT 10, 15; IN KEY $  
20 GO TO 10
```

(מחק את פקודת ה-BEEP)

בשלב זה המחשב מתרוצץ בין שורה 10 לבין שורה 20 ובחזרה, בלולאה אינסופית.

כעת אנחנו רוצים שמקש אחד יהיה שונה מכל המקשים. נבחר את המקש "0", למשל - הוא יהיה ההדק שלך: כלומר, התוכנית עובדת כרגיל ומדפיסה במרכז המסך את המקש עליו אתה לוחץ.

אבל - אם תלחץ על "0" (ההדק) אז ייצא "טיל" (*) מתחתית המסך, יעוף כלפי מעלה, יפגע באות הכתובה במרכז המסך ויעלם בשמים הכחולים



התוכנית תראה כך:

10 PRINT AT 10, 15 ; IN KEY \$

15 אם מקש "0" לחוץ - אז לך לשורה 130

20 GO TO 10

30 צייר טיל (*) היוצא
40 מתחתית המסך (15, 21)
· ועולה למעלה עד 15, 0
·
·

וכך עובדת התוכנית:

כל עוד לא לחצת על ההדק (מקש "0"), המחשב מתעלם משורה 15, מתרוצץ בין 0 לבין 20 ומדפיס במרכז המסך כל מקש עליו אתה לוחץ.

ברגע שתלחץ על "0" - המחשב קופץ לשורה 130 ומשגר את הטיל על המסך

איך כותבים את שורה 15 בשפת-מחשב? הנה כך:

15 IF INKEY\$ = "0" THEN GO TO 30

שפירושה: אם (IF) "0" לחוץ (IN KEY \$="0") - אז (THEN) לך לשורה 130

ואם לא? - אם "0" לא לחוץ?
במקרה כזה המחשב פשוט ממשיך לשורה הבאה בתור (שורה 20).

השלם את התוכנית במחשב שלך והרץ אותה:

15 IF INKEY\$ = "0" THEN GO TO 30

↑
(שים לב למראות
סביב ה-0)

□	□	□	□	□	□	□	□	□	□
□	□	□	□	□	□	■	□	□	□
□	□	□	□	■	□	□	□	□	□
□	□	□	□	□	□	□	□	□	□

כדאי שתוסיף פקודת BEEP בין 10 ל-20 כדי להשתכנע שהמחשב

"מתרוצץ" כל הזמן בין 10 ל-20, גם כאשר אינך לוחץ על

שום מקש.

כתוב גם, משורה 30 ואילך, פקודות שיציירו טיל שעף כלפי

מעלה על המסך.

בדוק את פעולת התוכנית: לחץ על מקשים שונים. לחץ "0".

תשובה 5 בעמוד 41

שים לב: ה-"0" חייב לבוא בין מראות. מדוע? - זאת תבין

בהמשך היחידה.

שכלל את התוכנית כך, שהיא לא תסתיים אחרי הטיל הראשון,

אלא תוכל לירות טיל בכל פעם שתלחץ על המקש "0".

תשובה 6 בעמוד 42

? מה יקרה אם תוסיף לתוכנית את השורה הבאה:

17 IF 2=3 THEN PRINT "MOSHE"

תשובה 7 בעמוד 42

קעת אתה יודע כל מה שצריך לדעת על-מנת לתכנת את משחק

הטלויזיה שתארנו בראשית היחידה.

אבל - כיון שפקודת ה-... THEN ... IF כל כך חשובה בשפת

התיכנות, הבה נתרגל אותה עוד קצת:

מה יקרה, לדעתך, אם תוסיף לתוכנית את השורה הבאה:

18 IF INKEY\$ = "Z" THEN NEW

הכנס את שורה 18 לתוכנית, הרץ אותה, לחץ על מקשים, ירה
 "טילים" - ולבסוף לחץ (בזהירות...) על המקש "Z!"
 האם הינך מופתע מן התוצאה?
 אם אינך מבין מה קרה - תשובה 8 בעמוד 42 תעזור לך.

? האם אתה זוכר מה מבצעת התוכנית הבאה:

```
10 FOR N=3 TO 10
20 PRINT AT N, N ; N
30 NEXT N
40 PRINT "SOF"
```

אם לא - הכנס אותה למחשב - והרץ.

כתוב תוכנית, המבצעת בדיוק אותו דבר על המסך - בלי
 להשתמש ב-FOR ... NEXT!
 רמז: העזר ב-... THEN ... IF (וגם LET ו-GO TO יבואו
 לעזרתך).


בדוק את התוכנית שלך על המחשב לפני שתשווה אותה ל:

תשובה 9 בעמוד 42

הערה: מאחר שפקודת IF-THEN היא פקודה על תנאי משתמשים הרבה
 פעמים בסימני היחס: > (גדול), < (קטן), <> (שונה)
 כדי להגדיר את התנאי. לדוגמא:
 IF X > 10 THEN... (X הוא משתנה)
 IF X < 10 THEN...
 IF X <> 10 THEN... (אם X שונה מ-10 אז...)
 (ראה את התשובה לשאלה הקודמת.)

הגיע הזמן לחזור אל משחק הטלויזיה שלנו

נזכיר לך את כללי המשחק:




חללית "הרעים" ($>$) חולפת ללא הרף לרוחב המסך משמאל לימין.

(FOR X = ... NEXT X)

החללית שלך (\$) אורבת לה למטה:

מקם את החללית שלך ב-(15, 21).

1




ברגע שנדמה לך ש"הרעים" נמצאים בדיוק מעל לחללית שלך, אתה לוחץ

על אחד המקשים של המחשב, שבחרת

בו להיות "הדק"

(IF INKEY\$ = ... THEN GO TO ...)

2



ברגע זה נעצרת חללית "הרעים",

והטיל שירית (*) מתחיל לעלות מן

החללית שלך ישר כלפי מעלה!

(FOR Y = ... NEXT Y)

3

מטרת המשחק: לפגוע בחללית האויב!

בעזרת הרמזים שפיזרנו בסוגריים, ובעזרת היידע שצברת,

אתה יכול לגשת לבנית התוכנית:

• בנה את התוכנית בשלבים: כתוב את שלב 1 והרץ אותו.

לאחר מכן הוסף את שלב 2 והרץ, וכן הלאה.

• גלה "בגים" בתוכנית שלך ופתור אותם.

הערה: יתכן שהתוכנית שאתה תכתוב תהיה שונה מזאת שנציע

לך כפתרון. לא נורא - השווה בין שתי התוכניות וראה איזה

יותר קצרה. בדרך כלל אפשר לכתוב כמה תוכניות המבצעות

אותה משימה. (כפי שראית בעמוד הקודם)

לאחר שכתבת את התוכנית והרצת אותה יתכן שתמצא את ה"בגיים"
הבאים:

- הטיל מוחק את החללית שלך - החל להעיף אותו משורה גבוהה יותר.

- עם הלחיצה על ההדק נעלמת חללית האויב - העבר את פקודת "IF IN KEY \$..." למקום מתוחכם יותר...

הכנס שיפור כזה בתוכנית, שהמשחק יימשך גם אחרי שהטיל ייעלם.

בג אפשרי: חללית האויב מהירי הקודם (זאת שלא פגעת בה) נשארת על המסך ומפריעה - CLS במקום המתאים יפתור זאת.

התוכנית המלאה בתשובה 10 ובעמוד 43

אגב - תהנה יותר אם תסתכל בתשובה רק אחרי התמודדות רצינית עם הבעיות.

משכללים את התוכנית

מתכנת חרוץ מנסה ללא הרף לשפר ולשכלל את התוכנית שהוא כותבו הבה נשכלל גם אנחנו את התוכנית.

להזכירך: מחשב הספקטרום עשיר בצבעים וצלילים - נצל זאת להנאתך!


הכנס לתוכנית פקודות כאלו, שיגרמו למחשב להדפיס על המסך, לאחר כל ירי טיל, את מספר הטילים שירית מתחילת המשחק.

רמזים:

- אתה זקוק למונה $(N=N+1)$ - הזכר בשעון הדיגיטלי.
הכנס אותו למקום מתאים בתוכנית. הרץ ובדוק אם אכן הוא סופר את הטילים שירית.
- אל תשכח לאפס את המונה בתחילת התוכנית.
- הדפס את המונה על המסך במקום כזה, שלא יפריע למשחק.
תשובה 11 בעמוד 43

שכלול נוסף

נמאס לך לספור בעצמך את מספר הפגיעות שלך במטרה?
המחשב יעשה זאת עבורך בשמחה! (רק תגיד לו איך...)
הבה נראה מה זאת אומרת, במונחי-מחשב, "פגיעה במטרה":



אם תחשוב ותענה על השאלות הבאות - תדע איך לפקוד על המחשב לספור את פגיעותיך במטרה:

? ירית טיל, והוא עולה למעלה - באיזו עמודה (x) צריכה להיות חללית האויב על-מנת שהטיל יפגע בה? $(15, 0, 5)$ _____
(לא זוכר מהי עמודה - ראה יחידה 2 עמוד 33)

? בניח שבחרת (כמונו) במשתנה M כמונה לפגיעות במטרה.
איזו מהפקודות הבאות תתאים לך:

- LET M = M+1 (א)
IF X=15 THEN LET M=M+1 (ב)
IF Y=15 THEN LET M=M+1 (ג)

תשובה 12 בעמוד 43

לא נותר לך אלא לאפס את מונה הפגיעות, להכניס את פקודת ה-"IF" במקום הנכון, ולפקוד על המחשב להדפיס את M במקום נוח על המסך.

הכנס את השיכלול לתוכנית ובדוק אם היא פועלת לשביעות רצונך.

תשובה 13 בעמוד 44

פקוד על המחשב לעצור את ריצת התוכנית (STOP) אחרי ירי של 10 טילים.

```

□ □ □ □ □ □ □ □ □ □
□ □ □ □ □ □ □ □ □ □
■ □ □ □ □ □ □ □ □ □
□ □ □ □ □ □ □ □ □ □

```

STOP - סטופ - עצור

תשובה 14 בעמוד 44

פקוד על המחשב להדפיס את אחוזי הפגיעה לפני שהוא עוצר.

תשובה 15 בעמוד 44

ולסיום המשחק:


כשתרצה למחוק את התוכנית הזאת מן הזיכרון - אל תכתוב "סתם" NEW: הכנס שינוי כזה שאם תפגע בחללית האויב - 10 פגיעות מתוך 10 נסיונות - התוכנית כולה תושמד!

פרק ב

עפרון וידיאו – צייר כאוות נפשך


אתה עומד לבנות תוכנית שתהפוך את המחשב לעיפרון-וידיאו ובעזרת לחיצה על מקשי המחשב המתאימים תוכל לצייר כאוות נפשך על מסך הטלויזיה שלך:

מחק כל תוכנית מזיכרון המחשב (NEW).
הבה נגדיר בדיוק מה צריכה התוכנית לבצע:




לחצת RUN - הופיעה הסיפורה
ø במרכז המסך:

1



בלחיצה ⇨ (ללא SHIFT):
הסיפורה ø נראית זזה ימינה
ומותירה אחריה שורה של
מספרי ø:

2



בלחיצת ⌂: הטור של ספרות
ה-ø עולה למעלה. אגב,
הספרות מתקדמות בכיוון החץ
כל זמן שתמשיך ללחוץ על המקשו

3

וכן הלאה לגבי החיצים שמאלה ולמטה. ⇨ ⇩

הערה: בחרנו את המקשים עליהם מצויירים החיצים האדומים רק מטעמי נוחות. באותה מידה יכולנו לבחור כל 4 מקשים אחרים - נקרא להם מקשי העפרון.

למעשה יש לך כל הידע הדרוש כדי לכתוב תוכנית כזאת. נסה
לכתוב אותה בלי להמשיך הלאה אל הרמזים שלנו.


רמז ראשון:

התוכנית בנויה סביב שני משתנים X ו-Y בפקודה:

PRINT AT Y, X ; Ø

כאשר תלחץ על אחד ממקשי העפרון, X ו-Y ישנו את ערכם
בהתאם למקש הלחוץ, והמחשב ידפיס Ø במקום המתאים.
(זאת בעזרת הפקודות IF... THEN הגורמות ל-Y ול-X
להשתנות בהתאם למקש הלחוץ.)

רמז שני:

אם המקש הלחוץ (IN KEY \$) הוא "8" (עליו מצויר )
אז הגדל את X ב-1 (בוע ימינה) וחזור לפקודת ה-PRINT.

רמז שלישי ואחרון:

	1Ø	_____
	2Ø	_____
	3Ø	PRINT AT Y, X; Ø
4 המקשים	{	4Ø IF INKEY\$ = "8" THEN LET X=X+1
		5Ø _____
		6Ø _____
		7Ø _____
עם	{	10Ø _____
החיצים		

התוכנית המלאה בתשובה 16 בעמוד 44

נסה "לצאת מן המסך" בעזרת העפרון: מה קורה? אם תנסה לצאת
ימינה או למטה - התוכנית תעצר. אם תנסה לצאת שמאלה או
למעלה - העפרון יחליף באופן אוטומטי את כיוונו בהמשך
נתגבר על בעיה זו.

וכמו תמיד, כך גם לגבי התוכנית הזאת:

משכללים את התוכנית

שנה את התוכנית כך, שחוד העפרון יהיה הסיפרה 2 -

(במקום 0).

כעת נכניס שכלול בתוכנית:

לפני תחילת הציור, המחשב ישאל אותך איזו סיפרה תהיה

חוד העפרון.

לשם כך עליך ללמוד פקודה חדשה:

INPUT (אינפוט) הכנס לתוכנית

ראשית, עליך להחליף את הסיפרה 2, הכתובה בפקודת ה-PRINT

בתוכנית שלך, במשתנה:

PRINT AT Y, X; A ←

ככה! אל תריץ עדיין את התוכנית. המחשב אינו יודע מה

הערך של A (הרץ ותיווכח - הודעת מחשב מס' 2 תופיע על

המסך).

ממש לפני פקודת ה-PRINT הכנס את הפקודה:

INPUT A (אינפוט)

```
□□□□□□□□□□
□□□□□□□■□□
□□□□□□□□□□
□□□□□□□□□□
```

שפירושה: הכנס לתוכנית את

הערך של המשתנה A

(תשובה 17 בעמוד 45)

הרץ את התוכנית.

מופיע [L] בפינה השמאלית-התחתונה של המסך.

המחשב הגיע לפקודה INPUT A והוא נעצר ומחכה שתגיד לו

לאיזה מספר אתה רוצה ש-A יהיה שווה!

בחר סיפרה כרצונך, כתוב אותה ולחץ ENTER.
עכשיו המחשב יודע כי A שווה לסיפרה שלחצת.

התחל לצייר בעזרת סיפרה זו על המסך.

רוצה להחליף עפרון?

עצור את ריצת התוכנית (BREAK), הרץ אותה שוב (RUN)
והכנס סיפרה חדשה.

נסכם:

בכל פעם שהמחשב מגיע בתוכנית לפקודת INPUT A, הוא עוצר
ומסמן ב- [L] כי הוא מחכה שתגיד לו מה הערך של A הכתוב
בפקודת ה-INPUT (אינפוט).

שכלול נוסף

שנה את "חוד העפרון" (לשנות סיפרה) מבלי למחוק את
הציור הקיים

הכנס לתוכנית פקודה בוספת: "אם מקש "P" (למשל) לחוץ,
אז לך לפקודת ה-INPUT!"

תשובה 18 בעמוד 45

הרץ את התוכנית.

בכל פעם שתמצא "להחליף עפרון" - לחץ "P". יופיע סימן [L]
למטה משמאל, האומר לך שהתוכנית מחכה ב-INPUT. הכנס
סיפרה ולחץ ENTER
כעת אתה יכול להמשיך ולצייר בעזרת עפרון החדש...

עפרון מוסיקלי...

משימה: הכנס לתוכנית שורה כזאת, שכאשר X או Y גדלים -
נשמע צליל הולך ועולה, וכאשר X או Y קטנים - יישמע
צליל הולך ויורדו
שים לבו כאשר אינך לוחץ כלל על המקשים - אסור שישמע צליל!

תשובה 19 בעמוד 45

שכלול מדהים

נמאס לך לצייר בעזרת ספרות. אתה רוצה להשתמש בכל
המקשים: ספרות, אותיות אנגליות ואפילו סימנים גרפיים!

כשהתוכנית בהמתנה ל-INPUT נסה להקיש אות אנגלית כלשהי
במקום סיפורה.

המחשב מסרב לקבל זאת, והוא גם "זרק" אותך, כעונש, מן
התוכנית... (הודעה 2)

מדוע אי-אפשר, בשלב זה, להכניס ב-INPUT אות אנגלית
כלשהי?

ובכן, כשהמחשב עוצר בפקודת INPUT A, הוא מחכה שתגיד לו
מה ערכו של ____ (P, Ø, A).

המשתנה A יכול להיות רק מספר ולא אות אנגלית ולא סימן
גרפי (A הוא משתנה-מספרי)

ולכן "התרגז" המחשב כאשר הוא חיכה ב-INPUT A למספר ואתה
הכנסת אות אנגלית!

משתנה מחרוזת (STRING VARIABLE)

השיכלול המדהים שתבצע עכשיו בתוכנית יעזור לך (ולמחשב)
להתגבר על "בעיה" זו:

בשתי השורות בהן מופיע A הוסף את הסימן \$ מיד אחרי האות

INPUT A\$:A

PRINT... A\$

כדי להריץ במחשב רק את 3 השורות 500, 510 ו-520.
כתוב - RUN 500 ולחץ ENTER. פקודה זו אומרת למחשב
להתחיל לרוץ מפקודה 500 ואילך.
סיימת? מחק את 3 השורות.

נסכם:

בתוכנית העפרון רצינו לצייר בעזרת כל סימן שנבחר. לכן
בפקודה:

```
PRINT AT Y, X ; A$
```

השתמשנו במשתנה "מחרוזת" (A\$) היכול לקבל כל צורה וסימן
שנרצה, או צירוף של מספר סימנים וצורות יחד (מחרוזת),
ולא רק מספרים, כמו המשתנה המספרי (A).

מה יקרה אם נשנה את השורה ל: ?

```
PRINT AT Y, X ; "A$"
```

נסה ותיווכח.

האם במקרה זה המחשב מתיחס ל-A\$ כאל מחרוזת או כאל משתנה
מחרוזת?

ועוד הערה:

שמו של משתנה מחרוזת חייב להיות אות אנגלית אחת
איזה ממשתני המחרוזת הבאים אינו חוקי? ?

A\$, BB\$, C3\$, 4\$, K\$

תשובה 22 בעמוד 45

עפרון צבעוני...

נחזור לעפרון הוידאו: שנה חזרה ל-

```
PRINT AT Y, X ; A$
```

.

הכנס שינוי כזה בתוכנית, שהמחשב ישאל אותך באיזה צבע
אתה רוצה לצייר. (וכמובן יצייר בצבע זה).

תשובה 23 בעמוד 45

שכלול למתכנת המתקדם:

כתוב עוד 4 פקודות IF ... THEN ... כאלו, שלא יאפשרו לעפרון ליפול מן המסך: בכל פעם שהעפרון יגיע לקצה המסך, ואתה תלחץ על חץ האומר לו להמשיך אל מחוץ למסך - "העפרון" יסרב למלא את הפקודה וישאר בקצה המסך!
תשובה 24 בעמוד 45


זהירות מוקשים

הגיע הזמן לעבור הלאה - אבל גם תוכנית כזאת חבל, סתם כך, לכתוב לה NEW!

הכנס לתוכנית פקודה כזאת, שאם, במהלך הציור, העפרון מגיע למשבצת 8, 10, למשל, - התוכנית משמידה את עצמה!
(כלומר, אתה טומן במשבצת 8, 10 מוקש - אם דורכים עליו התוכנית מושמדת (NEW)!)
לשם כך עליך ללמוד מושג חדש:

AND - וגם

אם $Y=8$ וגם $X=10$ אז השמד את התוכנית:
IF $Y=8$ AND $X=10$ THEN _____

AND (אֶנְד) - וגם נמצא ב: 

השלם את הפקודה, הכנס לתוכנית ונסה לעלות על המוקש...

פינת המשחק

אולי נתן לך המוקש רעיון למשחק "בשדה מוקשים"?
למשל: צריך להגיע מלמטה למעלה (למטמון) בלי לדרוך על מוקשים בלתי-נראים שפיזרת בתוכנית.
"פזר מוקשים", ותן לחברך לנסות ולהגיע למטמון!

פרק ג

מרוצי סוסים

המחשב כקוביה אלקטרונית

נתחיל בבניית תוכנית שתהפוך את המחשב לקוביה אלקטרונית:
בלחיצת כפתור הוא "יזרוק קוביה" ויראה לך איזה מספר עלה בגורלו


הרץ במחשב את התוכנית הבאה:

10 PRINT RND

20 GO TO 10

RND - (אַר. אָן. דִי): אומר למחשב לבחור מספרים באקראי.

(הגרלה של מספרים)



RND ← הדפס

T >
RAND

 ← אל תדפיס

בכל פעם שתבצע SCROLL (אינך זוכר איך? ראה עמוד 45
ביחידה 1) המחשב ימשיך ויראה לך מספרים נוספים. נסה.
בשלב זה אתה כבר יכול לפתוח "מפעל הגרלות" משלך...
אך נחזור לקוביה שלנו:

? כל המספרים שקיבלת יותר גדולים מ _____ (0.1, 0, 0.5)
ויותר קטנים מ _____ (1, 0.9, 2)

אגב: כשהמחשב כותב 0.6374. למשל, הכוונה היא ל-0.6374.

כלומר, המחשב בחר באקראי מספרים בין 0 לבין 1.
(RND לעולם לא תגדיל את המספר 1.0000 (אחד) עצמו)

שנה את התוכנית כך, שהמחשב יבחר מספרים בין 0 לבין 6:

10 PRINT RND * _____

20 GO TO 10

תשובה 25 בעמוד 46

נסה וראה אם אכן קיבלת מספרים בין 0 לבין 6.

INT (אינטג'ר - קיצור של INTEGER) - מספר השלמים
כדי לבנות "קוביה" ניפטר מן השברים המופיעים אחרי
הנקודה העשרונית:

4 . 23076682


וכך בקבל את ששת המספרים של הקוביה.

שנה את התוכנית שלך:

10 PRINT INT (RND * 6)

20 GO TO 10

סוגריים


 INTGER (אינטג'ר) פירושו
מספר השלמים.

פעולתו של ה-INT פשוטה מאוד:

מכל מספר הוא "לוקח" רק את מה שכתוב משמאל לנקודה
העשרונית - רק את מספר השלמים.

? מה שווה, לדעתך:

INT 7.99

תחילה חשוב, ואח"כ:

כתוב במחשב:

PRINT INT 7.99

ולחץ ENTER.

הרץ את התוכנית שבמחשב שוב ושוב וענה על השאלות הבאות:

? על המסך ודאי הופיע גם המספר - 0. האם מספר זה נמצא גם על קוביה אמיתית? _____ (לא/כן)

? בקוביה אמיתית מופיע גם המספר 6. האם הוא הופיע גם בקוביה האלקטרונית שלך? _____ (לא/כן)

כלומר, הקוביה האלקטרונית שלך "מגרילה" מספרים מ-0 עד 5, במקום מ-1 עד 6 כמו קוביה אמיתית.

שנה את התוכנית שלך כך, שתפעל כקוביה אמיתית:

10 PRINT INT (RND * 6) + ____

20 GO TO 10

תשובה 26 בעמוד 46

הרץ את התוכנית שוב ושוב ובדוק אם כל אחד מן המספרים:

6, 5, 4, 3, 2, 1 אכן הופיע על המסך לפחות פעם אחת.

כתוב תוכנית שתמלא עבורך טורים בטופס ה"טוטו":

0 - יהיה תיקו

1 - קבוצה מס' 1 מנצחת.

2 - קבוצה 2 מנצחת.


תשובה 27 בעמוד 46

כעת אפשר להשתעשע מעט

כתוב תוכנית כזאת, שבכל פעם שתלחץ על מקש כלשהו (בניח L) המחשב יציג על המסך תוצאה של זריקת 2 קוביות:

הערה: מחק את כל המסך (CLS)

בין זריקה לזריקה.



(הוסף צבע כך שכל מספר:

יבוא על נייר שונה:

PAPER 6 עבור 6 וכך הלאה)

שנה את התוכנית כך, שלפני "זריקת" כל קוביה ישמיע המחשב צליל שגובהו תלוי בתוצאה שמראה "הקוביה"!

תשובה 28 בעמוד 46

חגיגה על המסך

הוסף שורה אחת כזאת, שאם שתי הקוביות מראות על אותו מספר (למשל 3:3 או 2:2) אז מתחילה חגיגה קטנה: נשמעים צלילים וה-BORDER מחליף צבעים!

רמז: בתוך השורה הדרושה העזר בנקודותיים (:).

תשובה 29 בעמוד 46

? מה תבצע, לדעתך, התוכנית הבאה:


```
10 PRINT AT INT (RND*22), INT (RND*32) ; " ■ "
```

```
20 GO TO 10
```


לאחר שחשבת, הכנס למחשב והרץ.

שנה את התוכנית כך, שכל ריבוע יהיה בצבע אקראי וישמיע צליל אקראי.
תשובה 30 בעמוד 46

פינת המשחק – צוללות



1 המחשב מדפיס על המסך את הצוללת שלך (אאאאא - או כל צורה אחרת שתבחר) ושואל אותך (INPUT) כמה פצצות עומק לזרוק על הצוללת.



2 נניח שבחרת ב-40 פצצות. המחשב מתחיל לזרוק פצצות עומק (■) על פני המסך, בצורה אקראית (RND), עד שהמסך מתמלא ב-40 פצצות עומק.

אם הצוללת שלך (אאאאא) לא נפגעה - קיבלת 40 נקודות. אם היא נפגעה (אאא■א) - לא קיבלת נקודות כלל (או ירדו לך 20).

אתה יכול לשחק נגד חברך (כל אחד בתורו) או נגד המחשב - ואז הוא בוחר לעצמו, באקראי, את מספר הפצצות שהוא מוכן להסתכן בהן.


תוכנית המשחק בתשובה 31 בעמוד 47.

פינת הדיסקו

מתכוונן בקרוב לערוך מסיבת ריקודים בביתך? מדוע לא נהפוך אותה למסיבת-מחשבים?!
כתוב תוכנית כזאת, שתחליף צבעי - BORDER ומסך באקראי (RND) בערך בקצב (PAUSE) של מוסיקת דיסקו.
אל תשכח להזמין גם אותנו...


מירוצי-סוסים

נצא קצת אל האויר הפתוח, אל מגרש מירוצי-הסוסים:



1, 2 ו-3 הם שלושה סוסים היוצאים למרוץ מצד שמאל של המסך. מי שיגיע ראשון לצד ימין של המסך - מנצח.

1



המחשב "מגריל" מספר מתוך 1, 2 או 3. אם עלה מספר 3, למשל, בגורל, המחשב מוסיף לסוס מספר 3 עוד סיפורה בשורה שלו:

2




סוס מספר 1 ניצח! הוא הגיע ראשון לצד ימין כי המספר שלו עלה בגורל יותר פעמים מן הסוסים האחרים.

3

ושוב - הידע שצברת עד כה מספיק כדי לכתוב תוכנית כזאת!

כתוב תוכנית שתבצע את מירוצי-הסוסים.

אם אתה בכל זאת מתקשה - יש רמז בעמוד הבא.




נסה, בהתחלה, לכתוב תוכנית של מרוץ בו משתתף רק סוס אחד. הסוס הזה מתקדם רק אם המחשב "ימגיל" את המספר 1 מתוך שלושה מספרים: 1, 2 ו-3.

תשובה 32 בעמוד 47

עכשיו כתוב את התוכנית המלאה לשלושה סוסים!

תשובה 33 בעמוד 47


הצעות לשכלולים



צייר על המסך קו-גמר אליו צריכים הסוסים להגיע: בדוגמא שלנו ניצח סוס 2 כי הוא דרך על קו-הגמר.

תשובה 34 בעמוד 48

אמור למחשב לעצור את המרוץ אם סוס 1 או סוס 2 או סוס 3 לשם כך השתמש ב:



OR (או) - או

IF X1 = ____ OR X2 = ____ OR X3 = ____ THEN STOP

תשובה 35 בעמוד 48

הוסף פקודות כאלה שיגרמו למחשב לשאול אותך לפני המרוץ על איזה סוס אתה מהמר. (INPUT)

תשובה 36 בעמוד 48

קיצור התוכנית בעזרת DIM

תוכנית מירוצי-הסוסים כללה כ-18 פקודות. אפשר לשפר את התוכנית באופן משמעותי, להוסיף עוד סוסים - ובכל זאת לקצר את התוכנית למחצית אורכה!


כיצד?

לשם כך עליך ללמוד את הפקודה DIM (דימ)
 DIM הוא קיצור למלה DIMENSION - מימד)

מחק כל תוכנית קודמת מן המחשב.

הכנס את הפקודה הבאה:

10 DIM X(3)
 (הקפד לרשום
 סוגריים)



הפקודה הזאת החליפה בבת-אחת שלוש פקודות בתוכנית
 מירוצי-הסוסים:

10 LET X1 = 0
 20 LET X2 = 0
 30 LET X3 = 0

כאשר המחשב קיבל את הפקודה - DIM X (3) הוא ביצע את
 הפעולות הבאות:

- המחשב היקצה בזכרונו מקום לשלושה משתנים:
 X(1), X(2), X(3).
- נתן לכל אחד מהם ערך התחלתי - 0.

קבוצה כזאת של משתנים נקראת מערך של משתנים.
 לכולם יש אותו "שם משפחה" - X
 ולכל אחד מהם יש "שם פרטי" (מספר סדורי) שונה:



10 DIM X(3) : הרץ את התוכנית

: אמור למחשב להדפיס את ערכו של X(1)

PRINT X(1) ואח"כ ENTER.

? מה רשם המחשב בחלק העליון של המסך? _____ (1, 0)

באותו אופן בקש מהמחשב להדפיס את ערכם של שאר בני המשפחה: $X(2)$ ו- $X(3)$.

? מה יקרה אם תשאל אותו על $X(4)$?
עשה זאת!

הודעת מחשב 3

ביקשת מהמחשב להדפיס את ערכו של איבר רביעי, $X(4)$, במערך של שלושה איברים $DIM X(3)$ בלבד. המחשב לא היקצה בזכרונו מקום לאיבר הרביעי והוא מודיע לך על כך בהודעה מספר 3.

? האם המחשב מכיר את $X2$ (בלי סוגריים)? נסה!
אתה רואה כי המחשב אינו מכיר במשתנה כזה! (הודעה 2) -
איברי המערך חייבים לבוא עם סוגריים.

? מה, לדעתך עושה התוכנית הבאה:

```
10 DIM A(4)
20 FOR N=1 TO 4
30 INPUT A(N)
40 NEXT N
```

הרץ את התוכנית. המחשב מבקש ממך את ערכו של $A(1)$.
בחר מספר כלשהו ולחץ ENTER עכשיו יבקש המחשב ממך את ערכו של $A(2)$ וכך הלאה.

שים לב לשימוש שעשינו במשתנה N , כדי לקרוא לכל אחד מאיברי המערך!

תן למחשב הוראה להדפיס את $A(1)$ - האם זה הערך שנתת לו?

ועכשיו הוסף עוד 3 שורות לתוכנית כך, שאחרי "טעינת"
כל איברי המערך במספרים - המחשב יציג לך אותם אוטומטית
על המסך, בצורה הבאה:

A(1) = (הערך שהכנסת למשתנה הראשון)
A(2) = (הערך שהכנסת למשתנה השני)
A(3) = (הערך שהכנסת למשתנה השלישי)
A(4) = (הערך שהכנסת למשתנה הרביעי)

מתקשה? פנה לתשובה 37 בעמוד 48

לאחר שהרצת -

מחק את התוכנית והמשך...

הרץ את התוכנית הבאה:

```
10 DIM C(3)
20 LET N=INT (RND*3)+1
30 LET C(N)=5
40 PRINT N
```

? מי מהמשתנים C(1), C(2), C(3) שווה ל-5, ומי מהם שווה
ל-5?

כדי להשתכנע שאתה צודק אמור למחשב להדפיס את ערכי
C(1), C(2) ו-C(3).

הסבר: בניח כי תוצאת ההגרלה בשורה 20 היתה - 2.
המחשב הגיע לשורה 30 ונתן ל-C(2) את הערך - 5.
שני המשתנים האחרים נשארו - 0.

חזור והרץ את התוכנית ותיווכח לדעת שהמחשב בותן לאחד
המשתנים את הערך - 5 בהתאם לתוצאות ההגרלה.
(בהתאם לערכו של N).

? מה תעשה התוכנית אם בחליף את הפקודות ל:

30 LET C(N) = C(N)+5

35 BEEP 0.1, 6

40 GO TO 20

הרץ את התוכנית.

האם אתה רואה משהו על המסך?

האם התוכנית רצה כל הזמן?

ובכן, המחשב לא נח לרגע (הצלילים מעידים על כך): בשורה 20
הוא מגריל בין 1, 2 ו-3. (נניח שיצא 3, אז בשורה 30 הוא
מגדיל את C(3) ב-5 ורץ חזרה להגרלה.)

עצור את ריצת התוכנית (BREAK).

? האם, לדעתך, ערכיהם של 3 המשתנים C(1), C(2) ו-C(3)

יהיו שווים או שונים?

בקש מן המחשב להציג את ערכיהם של 3 המשתנים.

? מה יקרה לתוכנית אם תוסיף את השורה:

37 PRINT AT N*3, 10 ; C(N)

לאחר שחשבת...

הכנס את השורה והרץ.


עכשיו יש לך כל הכלים לשכלל את תוכנית מירוצי-הסוסים.
קצר אותה כמעט למחצית אורכה בעזרת DIM, ותוכל גם להגדיל
את מספר הסוסים ל-6, למשל.

תשובה 38 בעמוד 48


פרק ד

מחפשים את המטמון – משחק מסכם


במשחק שנפתח עכשיו, תצטרך לגייס את כל הידע שצברת לאורך 3 החוברות:



1 שדה המשחק מוגבל לחלק השמאלי-עליון של המסך. עם תחילת המשחק המחשב "מפזר" שם 5 מוקשים (*) באופן אקראי (RND). אחרי כ-3 שניות המוקשים נעלמים (CLS)



2 אחרי שהמוקשים נעלמו מן המסך (אך לא מזיכרוננו של המחשב), הופיע מחפש-האוצרות (■), והופיע המטמון (\$). משימתך: להביא את מחפש-האוצרות אל המטמון בלי לעלות על המוקשים הבלתי-נראים!



3 מחפש-האוצרות יכול ללכת רק ימינה (◀) או למטה (⬇️) (עפרון-וידיאו). אם הוא עולה על מוקש - המחשב כותב על המסך, ליד המוקש את המלה "BOOM". ראה על כריכת החוברת!

במהלך 3 החוברות עליהן עברת, צברת את כל הידע הדרוש על-מנת לכתוב את התוכנית של המשחק הזה. נסה לבצע זאת בכוחות עצמך. בדוק כל שלב על-ידי הרצתו במחשב. מצא בגימ ופתור אותם. הערה: ניתן לכתוב תוכנית מסורבלת בלי DIM. אנו ממליצים לך לכתוב את התוכנית עם DIM. אם אתה מתקשה פנה לרמזים בעמוד הבא.

רמז 1

הבעיה העיקרית של המשחק היא, לפזר 5 מוקשים ולשמור אותם בזיכרון כך, שכאשר מחפש-האוצרות יעלה על אחד מהם, המחשב ידע זאת וידפיס "BOOM!"

<u>DIM Y(5)</u>	<u>DIM X(5)</u>	לשם כך צריך להשתמש בשני מערכים:
Y(1) ,	X(1)	נביח שהמוקש הראשון נמצא ב-
Y(2) ,	X(2)	ואז המוקש השני נמצא ב-
Y(3) ,	X(3)	השלישי נמצא ב-
Y(4) ,	X(4)	הרביעי נמצא ב-
Y(5) ,	X(5)	החמישי נמצא ב-

רמז 2

מה עושה התוכנית הבאה:

```
10 DIM Y(3)
20 DIM X(3)
30 FOR N=1 TO 3
40 LET Y(N) = INT (RND*5)
50 LET X(N) = INT (RND*5)
60 PRINT AT Y(N), X(N); "*"
70 NEXT N
```

רמז 3


כדי שהמחשב ידפיס "BOOM" כשמחפש-האוצרות עולה על מוקש אל תשכח שבפקודת ה-IF אתה חייב להשתמש ב-AND (ראה עמוד 25 בחוברת זו).

התוכנית המלאה בתשובה 39 עמוד 49

משכללים את המשחק

נסה להכניס את השיכלולים הבאים (הזהר ממגבלות הזיכרון):

- לפני שהמשחק מתחיל המחשב שואל את שמך. כאשר אתה מגיע למטמון הוא מדפיס: "יפה מאוד משה" או כל מחמאה אחרת שתבחר.
- עלית על מוקש? הוא מדפיס "BOOM" ומפסיק את המשחק.
- הגעת למטמון? המחשב מציג לך את המוקשים שהיו מפוזרים בשטח (והפעם לא מוחק אותם יותר, כדי שתראה איזה סכנות עברת...).
- המשחק קל מדי?
אחרי שהמוקשים נעלמים מופיע מעבר צר:



ומותר לעבור רק דרכו
(וגם להזהר מן המוקשים).

- עדיין קל מדי?
אחרי שהמוקשים נעלמים, המעבר מופיע בשורה אקראית!
- אולי המחשב יגיד לך גם כמה זמן לקח לך להגיע למטמון?

(השיכלולים שאנחנו הצלחנו
להכניס נמצאים בעמוד 50)

וכעת

קח צבע, קח צליל - וקשט בהם את התוכנית שלך.

ולסיום

הגעת לסופה של יחידת הלימוד השלישית והאחרונה בקורס. בתחילת הקורס הבטחנו לך כי תלמד להשתלט על המחשב בדרך פעלתנית ומהנה, ואנו מקוים שאכן כך עשית.

תוך כדי פיתוח הרעיונות למדת להכיר מושגים הכרחיים לכתיבת תוכניות בשפת BASIC. השתדלנו ללמד את החומר בצורה שתקל עליך ליישם את הידע שרכשת גם על מחשבים אחרים. דע לך כי לכל סוג של מחשב תכונות מיוחדות משלו, אך הבסיס לכולם - משותף.

בעתיד תוכל למצוא על המדפים חוברות נוספות של חברת "מחשבת" העוסקות במחשבים אחרים ובתוכניות מתקדמות יותר.

שאלת חזרה לסיכום

לפניך רשימת מושגים שלמדת ביחידה זו. עבור על כל אחד מהם ובדוק אם אתה זוכר אותם. (תוכל לרענן את זכרוןך בעזרת מספרי העמודים המופיעים בסוגריים ליד כל מושג):

(26)	RND	(7)	IN KEY \$
(27)	INT	(9)	IF... THEN
(12)	<>	(16)	STOP
(32)	OR	(19)	INPUT
(33)	DIM	(23)	מחרוזת (STRING)
(34)	הודעת מחשב 3	(24)	משתנה מחרוזת
		(25)	AND

תשובות

תשובה 1 מעמוד 8:

אמרנו, כי כאשר המחשב מגיע בתוכנית ל-IN KEY \$, הוא קורא את המקש הלחוץ באותו רגע וממשיך הלאה. אם אין מקש לחוץ, הוא מתעלם מההוראה וממשיך הלאה.

המסקנה: המחשב מתרוצץ כל הזמן משורה 10 ל-20 וחזרה ל-10 מבלי לעשות שום דבר. המחשב "מסתובב" על-ריק.

התבונן במסך - בחלק התחתון - אין שום הודעה של המחשב - זה סימן שהוא עובד. כמו כן, אם תלחץ על BREAK הוא יעצר ותופיע הודעה בתחתית המסך.

תשובה 2 מעמוד 8:

המחשב רץ כל הזמן משורה 10 ל-20 וחזרה ל-10. כל פעם שהוא עובר את 10 הוא מדפיס את המקש הלחוץ. לכן, ככל שתלחץ על מקש כלשהו זמן רב יותר, יופיע הסימן של המקש מספר גדול יותר של פעמים.

15 BEEP 0.1, 10

תשובה 3 מעמוד 8:

תשובה 4 מעמוד 9:

10 PRINT AT 10, 15; IN KEY \$

20 GO TO 10

תשובה 5 מעמוד 11:

יש להוסיף לתוכנית הקודמת את הפקודות:

15 IF INKEY\$ = "0" THEN GO TO 30

16 BEEP 0.2, 10

והטיל עף למעלה:

30 FOR Y=21 TO 0 STEP-1

35 PRINT AT Y, 15; "*"

37 PAUSE 5

40 PRINT AT Y, 15; "□"

50 NEXT Y

(שורה 16 באה רק כדי שתשתכנע כי המחשב מתרוצץ כל הזמן

בין 10 ל-20.)

תשובה 6 מעמוד 11:

השורה הנוספת:

60 GO TO 10

שים לב כי לאחר שהטיל נעלם לא מופיע שום דבר על המסך. רק אם תלחץ על מקש תקבל סימן במרכז המסך.

תשובה 7. מעמוד 11:

לא יקרה כלום!

מאחר שהתנאי $2=3$ אף פעם לא מתקיים, המחשב מתעלם משורה 17 וממשיך הלאה.

תשובה 8 מעמוד 12:

אתה בודאי זוכר כי הוראת NEW אומרת למחשב למחוק את התוכנית מהזיכרון. לכן, פקודה 18 היא שורה "השמדה-עצמית" של התוכנית: ברגע שאתה לוחץ על "Z" שורה 18 מתבצעת, וכל התוכנית במחקת מהזיכרון.

תשובה 9 מעמוד 12:

```
10 LET N=3
20 PRINT AT N, N; N
30 LET N=N+1
40 IF N>10 THEN GO TO 60
50 GO TO 20
60 PRINT "SOF"
```

אתה רואה כי התוכנית עם FOR-NEXT יותר משוכללת!

תשובה 10 מעמוד 14:

החללית אורבת	10 PRINT AT 21, 15; "\$"
	20 FOR X=0 TO 31
	[25 PRINT AT 5, X; ">"
	29 BEEP 0.1, X
הרעים טסים	28 IF INKEY\$ = "P" THEN GO TO 50
	30 PRINT AT 5, X; " □ "
	35 NEXT X
	40 GO TO 20
	[50 FOR Y=20 TO 0 STEP-1
	60 PRINT AT Y, 15; "*"
	63 BEEP 0.1, Y
הטיל עולה	65 PRINT AT Y, 15; " □ "
	70 NEXT Y
נקה את המסך	[80 CLS
ושגר עוד רעים	90 GO TO 10

(טים לב כיצד הכנסנו
את הצלילים בשורות
29 ו-63). את הצבעים אנו
משאירים לך.

תשובה 11 מעמוד 15:

בוסף את השורות הבאות לתוכנית הקודמת:

```
5 LET N=0
15 PRINT AT 0, 2; N
85 LET N=N+1
```

תשובה 12 מעמוד 15:

נבחר באפשרות השניה: כאשר X=15 אז נפגע ברעים:

```
IF X=15 THEN LET M=M+1
```

(אם המיקום של החללית שלך שונה משלנו אז שנה שורה זאת
בהתאם.)

תשובה 13 מעמוד 16:

הפקודות הנוספות שצריך להכניס לתוכנית הם:

```
4 LET M=0
16 PRINT AT 0, 30; M
87 IF X=15 THEN LET M=M+1
```

תשובה 14 מעמוד 16:

נוסיף את הפקודה:

```
75 IF N=10 THEN STOP
```

(אם אתה מגלה בעיות בזכרון אז הקטן את מספר היריות במטווח.)

תשובה 15 מעמוד 16:

```
75 IF N=10 THEN PRINT AT 0,15;
```

```
M/N * 100 : STOP
```

שים לב: אחרי THEN אפשר לכתוב מספר פקודות (המופרדות ע"י בקודותיים).
המחשב יבצע את כל הפקודות הללו רק אם תנאי ה-IF יתקיים!

תשובה 16 מעמוד 18:

```
10 LET X=15
20 LET Y=10
30 PRINT AT Y, X; 0
(ימינה →) 40 IF INKEY$="8" THEN LET X=X+1
(מעלה ↑) 50 IF INKEY$="7" THEN LET Y=Y-1
(מטה ↓) 60 IF INKEY$="6" THEN LET Y=Y+1
(שמאלה ←) 70 IF INKEY$="5" THEN LET X=X-1
100 GO TO 30 - פונה לשרטט
30 בפקודה
```

תשובה 17 מעמוד 19:

25 INPUT A

תשובה 18 מעמוד 20:

80 IF INKEY\$ = "P" THEN GO TO 30

תשובה 19 מעמוד 21:

83 IF IN KEY \$ = " " THEN GO TO 25

85 BEEP 0.1, X+Y

תשובה 20 מעמוד 22: 0.1 , -1.4 , 3

תשובה 21 מעמוד 23:

"4*7=" מחרוזת

4*7 ביטוי חשבוני

תשובה 22 מעמוד 24: BB\$, C3\$, 4\$ - אינם חוקיים.

תשובה 23 מעמוד 24:

27 INPUT "COLOUR ?"; B

30 PRINT INK B; AT Y, X; A\$

שים לב, בשורה 27, איך אפשר להדפיס על המסך את "שאלת ה-INPUT".

תשובה 24 מעמוד 25:

השורות הבאות לא נותנות לעפרון לברוח ימינה או שמאלה:

90 IF X>31 THEN LET X=31

92 IF X<0 THEN LET X=0

הוסף עוד שתי שורות כך, שהעפרון לא יברח למעלה ולמטה.

תשובה 25 מעמוד 27:

10 PRINT RND*6

יש להכפיל ב-6:

תשובה 26 מעמוד 28:

יש להוסיף 1 ואז יוגרלו המספרים: 1, 2, 3, 4, 5, 6:

10 PRINT INT (RND*6)+1

תשובה 27 מעמוד 28:

10 PRINT INT (RND*3)

20 GO TO 10

תשובה 28 מעמוד 29:

המחשב אינו ממשיך -
אלא אם כן תלחץ על מקש
כלשהו.

10 PAUSE 0

30 CLS

שים לב לשורה 10!

הגרלת הקוביות

40 LET N=INT (RND*6)+1

50 LET M=INT (RND*6)+1

הצגת הקוביות בצבע
והשמעת הצלילים.

60 BEEP 0.2, M*3 : PRINT PAPER M;
INK 9; AT 10, 10; M

70 BEEP 0.2, N*3 : PRINT PAPER N; INK 9;
AT 10, 20; N

80 GO TO 10

תשובה 29 מעמוד 29:

72 IF M=N THEN

הוסף את השורה:

FOR K=1 TO 7 : BORDER

K : BEEP 0.2, 10 : BEEP

0.2, 20 : BEEP 0.2, 8 : NEXT K

תשובה 30 מעמוד 30:

10 PRINT INK INT (RND*7);

AT INT (RND*22), INT (RND*32); " ■ "

15 BEEP 0.1, INT (RND*20)

20 GO TO 10

תשובה 31 מעמוד 30:

הדפס את הצוללת	10 PRINT AT 7, 10 ; "XXXXXX"	
שאל על כמה פצצות	[20 INPUT "KAMA BOMBOT ? <input type="text"/> <input type="text"/> <input type="text"/> "; M
רוצים להמר (M פצצות)		
זרוק M פצצות		
במקומות אקראיים		
על המסך	30 FOR N=1 TO M	40 PRINT AT INT (RND*22), INT (RND*32) ; " ■ "
	50 NEXT N	
הדפס את מסי הפצצות	60 PRINT AT 0, 0 ; M	

אפשר להוסיף ולשכלל עוד את המשחק ואת התוכנית - נסה!

תשובה 32 מעמוד 32:

```
5 LET X1=0
10 LET N=INT (RND*3)+1
20 IF N=1 THEN LET X1=X1+1
30 PRINT AT 3, X1 ; "1"
40 GO TO 10
```

תשובה 33 מעמוד 32:

	10 LET X1=0		
איפוס משתנים	20 LET X2=0		
	30 LET X3=0		
	40 LET N=INT (RND*3)+1		
המחשב מדפיס	[50 PRINT AT 3, X1; "1"	
את התקדמות			60 PRINT AT 6, X2; "2"
הסוסים			70 PRINT AT 9, X3; "3"
הגדל ב-1 את המשתנה	[80 IF N=1 THEN LET X1=X1+1	
של הסוס שעלה			90 IF N=2 THEN LET X2=X2+1
בגורל			100 IF N=3 THEN LET X3=X3+1
	110 GO TO 40		

שים לב! הפעם הקדמנו את פקודות ה-PRINT לפקודות ה-IF, וזאת כדי שלפני תחילת המירוץ יציג המחשב את שלושת הסוסים על קו הזינוק.

תשובה 34 מעמוד 32:

הפקודות לציור קו הגמר הן:

```
2 FOR K=2 TO 10
4 PRINT AT K, 20 ; " ■ "
6 NEXT K
```

תשובה 35 מעמוד 32:

כאמור נשתמש בהוראה OR (או) בפקודת IF כדי לעצור את התוכנית:

```
75 IF X1=20 OR X2=20 OR X3=20
    THEN STOP
```

תשובה 36 מעמוד 32:

```
34 INPUT "MI MAGIA RISHON?   "; A
36 PRINT A
```

תשובה 37 מעמוד 35:

הגדרת המערך	10 DIM A(4)
קולט את הערכים	[20 FOR N=1 TO 4 30 INPUT A(N) 40 NEXT N
של משתני	
המערך	
מדפיס את ערכי	[50 FOR N=1 TO 4 60 PRINT "A(";N;")=";A(N) 70 NEXT N
המשתנים	
של המערך	

תשובה 38 מעמוד 36:

תחרות של שישה סוסים: בציג רק את הפקודות שעוסקות במירוץ עצמו:

```
10 DIM X(6)
20 LET N=INT (RND*6)+1
30 LET X(N) = X(N)+1
35 BORDER N
40 PRINT PAPER N; INK 9; AT 3*N, X(N); N
50 GO TO 20
```

את כל השיכלולים מסביב נשאיר לך...

איפוס מיקום	[5 LET Y=0
העפרון	[7 LET X=0
הגדרת שני	[10 DIM Y(5)
המערכים	[20 DIM X(5)
טעינת שני	[30 FOR N=1 TO 5
המערכים	40 LET Y(N)=INT (RND*11)
בערכים אקראיים	50 LET X(N)=INT (RND*11)
והדפסת המוקשים	60 PRINT AT Y(N), X(N); "*"
על המסך	70 NEXT N
פסק-זמן	[80 PAUSE 100
מחיקת המוקשים	100 CLS
הדפסת מחפש-	[110 PRINT AT 10, 10; "\$"
האוצרות והאוצר	[120 PRINT AT Y, X; "■"
עפרון וידיאו	[130 IF INKEY\$ = "6" THEN LET Y=Y+1
	[140 IF INKEY\$ = "8" THEN LET X=X+1
בדיקת המיקום	[150 FOR N=1 TO 5
של העפרון -	160 IF Y(N)=Y AND X(N)=X
האם עלה על	THEN PRINT AT Y+1,X; "BOOM!" : STOP
מוקש?	[170 NEXT N
	180 GO TO 120

אנחנו הוספנו את השיכלולים הבאים:

איפוס השעון
מוקשים אדומים
דולר ירוק...
מעבר הכרחי

הדפסת השעון
וקידום השעון

הצגת המוקשים

```

3   LET A=0
60  PRINT INK 2; AT Y(N), X(N); "*"
110 PRINT INK 4; AT 10, 10; "$"
112 PRINT AT INT (RND*4)+4, 0; " ██████████ ██████████
██████████ "
120 PRINT INK 1; AT Y, X; " "
122 PRINT AT 0, 20; A : LET A=A+1
125 IF IN KEY $ = " " THEN GO TO 122
145 BEEP 0.2, 10
160 IF Y(N)=Y AND X(N)=X THEN PRINT AT Y+1,X;
    "BOOM!" : GO TO 200
175 IF Y=10 AND X=10 THEN PRINT AT 12, 20;
    "YOFI!" : GO TO 200
200 FOR N=1 TO 5
210 PRINT INK 2; FLASH 1; AT Y(N), X(N);
    "*"
220 NEXT N
    
```

שים לב לפקודת FLASH בשורה 210: תמצא את המקש ליד מקש ה-PAPER.

FLASH 1 - הדפסה מהבהבת.

FLASH 0 - מפסיק את ההבהוב.

עד כה יצאו לאור סדרות "מחשבת" על:

- מחשב COMMODORE 64
(3 החוברות הראשונות מתאימות גם ל-VIC-20).
- מחשבי סינקלר: ZX-SPECTRUM
ZX-81
- מחשבי APPLE
- מחשבי ATARI
- מחשב SPECTRAVIDEO
- מחשב AMSTRAD

משהו על יחידה 3:

ביחידת לימוד זאת הנקראת:

"פיתוח משחקי מחשב טלויזיה"

תעסוק בפיתוח תוכניות מורכבות יותר תוך

שימוש בפקודת התנאי: IF... THEN.

ביחידה זאת תשתמש כבר בכל הכלים

היסודיים הקיימים בשפת ה-BASIC, כדי לפתח

את המשחקים.

מחשבת מ.ל. בע"מ / מערכות למידה